


Introduction to NCL Graphics

Paneling Plots

Part IV in the series

November 18, 2014

Mary Haley



*Sponsored by the National
Science Foundation*

You may want to bookmark this link

http://www.ncl.ucar.edu/Training/Webinars/NCL_Graphics/PanelDemo/

NCL Home Page -> Intro to NCL Graphics Webinars -> Webinar #4

The screenshot shows the NCAR Command Language (NCL) website. At the top, there's a navigation bar with links for NCAR, CISL, VETS, NCL (selected), Examples, Functions, Resources, Popular Links, What's New, Support, External, Download, and Contributors. Below the navigation is the NCAR logo and a search bar with options for advanced search and Google Custom Search.


The main content area features several sections:

- What is NCL?**: Describes NCL as an interpreted language for scientific data analysis and visualization.
- How to Get NCL**: Notes it's portable, robust, and free, available as binaries or open source.
- Supported Data Formats**: Lists netCDF3/4, GRIB1/2, HDF-SDS, HDF4-EOS, binary, shapefiles, and ascii files.
- Built-in Functions**: Mentions numerous analysis functions.
- High Quality Graphics**: States that graphics are easily created and customized with hundreds of graphic resources.
- Example Scripts**: Indicates many example scripts and their corresponding graphics are available.

On the right side, there's a sidebar with sections for **Release Information**, **Announcements**, and **Citing NCL**. The **Announcements** section contains a message about the "Introduction to NCL Graphics" webinar being rescheduled for November 18 & 19, which is circled in black.

At the bottom, there's a footer with links to ©2014 UCAR, Privacy Policy, Terms of Use, Contact the Webmaster, and Sponsored by NSF.

Introduction to NCL Graphics



About this series

- Fourth in a series:
 - The basics and XY plots
 - Creating contours plots
 - Vector plots and overlays
 - Panel plots
 - Primitives
- Webinars are expanded versions of NCL workshop lectures . . . with more time for demos!
- To see other webinars in this series:

http://www.ncl.ucar.edu/Training/Webinars/NCL_Graphics/

My goals

- Get you comfortable with creating NCL graphics . . . won't make you an expert!
- Sneak in tips for more advanced users
- Answer your burning questions

Assumptions

Familiar with:

- basic NCL language features
- NCL array syntax
- reading data off a NetCDF file using NCL
- basic structure of an NCL graphics script
- NCL/NetCDF data model
- UNIX

Outline

- Panel plot gallery
- Three ways to panel plots
- Line-by-line scripts
- Demo
- Tips

Panel plot gallery

Many images are from the “panels” example page


NCL Home Page -> Examples

Browser search for “panel”


<http://www.ncl.ucar.edu/Applications/panel.shtml>

Panel plots – multiple plots on a page


OLR: Raw/Smooth Daily Climatology: 1994-1998



Variation of $T_t(0)$ with Heating Location and Rossby Radius




Anomalies: Daily OLR: Areal Averaged & Filtered: 20-100 days: nw=201




MJO graphic
Courtesy of Dennis
Shea/CGD

ESMF Regridding





Original EASE grid (721,721)



Regridded to 0.25 deg grid (359 x 1439)



Paneling bar plots, dummy data




█ first

second


third

fourth


Panel plot with two labelbars




Maps with filled and transparent bars



Zonal wind (m/2)





Zonal wind (m/2)




OLR file
from
Bob
Setzenfand




chi200_ud_smooth.nc




Meteogram for LGSA, 28/12Z




John Ertl, FNMO




Arrowtooth Flounder




Rock Sole




Thornyhead Rockfish




Sculpin



Pacific Cod




Snow Crab



The "Alaska Essential Fish Habitat Species" shapefile was downloaded from:

<http://alaskafisheries.noaa.gov/habitat/efh/efhshp/default.htm>


panels
with a
specially
created
labelbar



Time = 0

Time = 1

Two panel
plots in one



Zonal Wind



300 mb

Time = 0

Zonal Wind

300 mb

Time = 1



Outline


- Panel plot gallery
- Three ways to panel plots
- Line-by-line scripts
- Demo
- Tips

Paneling plots with NCL – three methods



1. Using “`gsn_panel`” procedure
2. Setting `vpXF` / `vpYF` / `vpWidthF` / `vpHeightF` resources
3. Using “`gsn_attach_plots`” function

Note: `gsn_panel` uses method #2 “under the hood”

Plot created using “gsn_panel”




Plot created using vpXXXXF resources



Courtesy Adam Phillips, NCAR CGD

Plot created using “gsn_attach_plots” and “gsn_panel”



Method 1: using “gsn_panel”

```
gsn_panel( wks, plots, (/n,m/), pres )
```

- *wks* is the workstation
- *plots* is a 1D array of plots to panel
- (/*n,m*/) is the configuration desired
(rows x columns)
- Special case for third argument:
of plots per row: (/*n1,n2,n3,...*/)
- *pres* is an optional resource list for
customizing behavior of gsn_panel

gsn_panel specifics

- Plots should be same size, or very close!
- Plots are drawn from left-to-right, top-to-bottom
- Can have common title, labelbar, figure captions
- Can have multiple sets of panels on one page

Outline

- Panel plot gallery
- Three ways to panel plots
- **Line-by-line scripts**
- Demo
- Tips

gsn_panel scripts

Scripts that follow are “slp1a.ncl” and “slp1b.ncl”

Data file also found here, “SLP.1979_2003.nc”

http://www.ncl.ucar.edu/Training/Webinars/NCL_Graphics/PanelDemo/

The screenshot shows a web page titled "Scripts and files for webinar paneling demos". The page header includes links for NCL, Examples, Functions, Resources, Popular Links, What's New, Support, and External. The main content area has sections for "Scripts referenced in panel lecture and demo" (listing slp1a.ncl through slp1i.ncl), "Controlling where on the NDC square to panel your plots" (listing slp_full_panel.ncl through slp_three_panels.ncl), and "Fixing issues with plots that are running off the page" (listing panel_too_big.ncl, panel_too_big_boxes.ncl, and panel_too_big_fix.ncl). At the bottom, there is a copyright notice: "©2014 UCAR | Privacy Policy | Terms of Use | Contact the Webmaster | Sponsored by NSF".

```
load "$NCARG_ROOT/lib/ncarg/nclscripts/csm/gsn_code.ncl"
load "$NCARG_ROOT/lib/ncarg/nclscripts/csm/gsn_csm.ncl"

;---Read "SLP" variable off NetCDF file
filename = "SLP.1979_2003.nc"
a        = addfile(filename,"r")
slp      = a->slp           ; time x lat x lon (300 x 73 x 144)

;---Open an X11 window
wks = gsn_open_wks("x11","slpla")

;---Set some plotting options
res          = True
res@cnFillOn = True           ; Turn on color fill

;---Create plots of first four time steps
plot0 = gsn_csm_contour_map(wks,slp(0,:,:),res)
plot1 = gsn_csm_contour_map(wks,slp(1,:,:),res)
plot2 = gsn_csm_contour_map(wks,slp(2,:,:),res)
plot3 = gsn_csm_contour_map(wks,slp(3,:,:),res)

;---Panel all four plots in a 2 x 2 configuration
pres = True
gsn_panel(wks, (/plot0,plot1,plot2,plot3/),(/2,2/),pres)
```

Same script, but with a “do loop”





```
;---Read "slp" variable off NetCDF file
filename = "SLP.1979_2003.nc"
a        = addfile(filename,"r")
slp      = a->slp                                ; 300 x 73 x 144

;---Open an X11 window
wks = gsn_open_wks("x11","slp1b")

;---Set some plotting options
res          = True
res@cnFillOn = True                               ; Turn on color fill

;---Loop across first four time steps and create a plot
nplots = 4
plots  = new(nplots,graphic) ; create array to hold plots
do i=0,nplots-1
  plots(i) = gsn_csm_contour_map(wks,slp(i,:,:),res)
end do

;---Panel all four plots in a 2 x 2 configuration
pres = True
gsn_panel(wks,plots,(/2,2/),pres)
```



Outline

- Panel plot gallery
- Three ways to panel plots
- Line-by-line script
- **Demo**
- Tips

Demo

Start with “slp1b.ncl” script from previous example

http://www.ncl.ucar.edu/Training/Webinars/NCL_Graphics/PanelDemo/


Outline

- Panel plot gallery
- Three ways to panel plots
- Line-by-line script
- Demo
- **Tips**


Method 2: setting “vpXXXXF” resources

- Use if plots are different sizes
- Use if you need more control over location and/or size of plots
- `vpXF` / `vpYF` / `vpWidthF` / `vpHeightF` are the resources you need
- Use “`drawNDCGrid`” to draw a handy “viewport” grid

Plot 1




Plot 2




Method 3: using “gsn_attach_plots”

Use if plots need to share a common axis



“panel_10.ncl”
on panel
examples
page

“xy_23.ncl”
on panel
examples
page



Tips for debugging panels gone bad

- Plots are not the same size, so paneling doesn't work
- Plots are going off the screen

```
;---Generate arrays to hold dummy data
npts      = 50
nplots    = 6
y         = new((/nplots,npts/),float)
ybeg     = (/0.00001,-5, 100000, 0, 10, 100/) ; begin values
yend     = (/1.00000, 5, 500000,20, 50, 200/) ; end values







wks = gsn_open_wks("x11","panel_too_big")

res          = True
res@gsnDraw  = False
res@gsnFrame = False
res@vpWidthF = 0.8           ; set width of plot
res@vpHeightF= 0.3           ; set height of plot

;---Create dummy data and plots in same loop
plot = new(nplots,graphic)
do i=0,nplots-1
  y(i,:) = random_uniform(ybeg(i),yend(i),npts)
  plot(i) = gsn_csm_y(wks,y(i,:),res)
end do

;---Panel these plots
pres = True
gsn_panel(wks,plot,(/3,2/),pres)
```

What's wrong with this set of plots?



```
;---Generate arrays to hold dummy data
npts      = 50
nplots    = 6
y         = new((/nplots,npts/),float)
ybeg     = (/0.00001,-5, 100000, 0, 10, 100/) ; begin values
yend     = (/1.00000, 5, 500000,20, 50, 200/) ; end values


wks = gsn_open_wks("x11","panel_too_big")

res          = True
res@gsnDraw  = False
res@gsnFrame = False
res@vpWidthF = 0.8           ; set width of plot
res@vpHeightF= 0.3           ; set height of plot

;---Create dummy data and plots in same loop
plot = new(nplots,graphic)
do i=0,nplots-1
  y(i,:) = random_uniform(ybeg(i),yend(i),npts)
  plot(i) = gsn_csm_y(wks,y(i,:),res)
end do

;---Panel these plots
pres          = True
pres@gsnPanelBoxes = True      ; draws red boxes around each plot
gsn_panel(wks,plot,(/3,2/),pres)
```

`pres@gsnPanelBoxes = True`



```

;---Generate arrays to hold dummy data
npts      = 50
nplots   = 6
y         = new((/nplots,npts/),float)
ybeg     = (/0.00001,-5, 100000, 0, 10, 100/) ; begin values
yend     = (/1.00000, 5, 500000,20, 50, 200/) ; end values

wks = gsn_open_wks("x11","panel_too_big")







res          = True
res@gsnDraw  = False
res@gsnFrame = False
res@vpWidthF = 0.8           ; set width of plot
res@vpHeightF= 0.3           ; set height of plot

;---Create dummy data and plots in same loop
plot = new(nplots,graphic)
do i=0,nplots-1
  y(i,:) = random_uniform(ybeg(i),yend(i),npts)
  plot(i) = gsn_csm_y(wks,y(i,:),res)
end do

;---Panel these plots
pres          = True
pres@gsnPanelScalePlotIndex = 2      ; base scaling on 3rd plot
gsn_panel(wks,plot,(/3,2/),pres)


```

pres@gsnPanelScalePlotIndex = 2



gsnPanelScalePlotIndex

Useful when plots have different tickmarks



"panel_23.ncl"
on panel
examples
page


Panel resources for debugging

- `res@gsnPanelScalePlotIndex = plot_index`
 - If one plot is larger than the others
- `res@gsnPanelDebug = True`
 - Prints debug information about size and location of paneled plots
- `res@gsnPanelBoxes = True`
 - Draws bounding boxes around each plot element so you can see true size
- `res@gsnPanelXF / res@gsnPanelYF`
 - Use these to force plots to line up
- `vpXF/vpYF/vpWidthF/vpHeightF`
 - If you need to abandon `gsn_panel!`


Multiple panels on one page

- Panels are drawn in a unit square
- By default:
 - `-gsnPanelTop` = 1.0
 - `-gsnPanelBottom` = 0.0
 - `-gsnPanelLeft` = 0.0
 - `-gsnPanelRight` = 1.0
- Can use these resources to restrict area for panels

gsnPanelTop



gsnPanelLeft



gsnPanelRight

gsnPanelBottom


Tip: NDC grid was drawn with drawNDCGrid

```
pres          = True
pres@gsnFrame      = False    ; Don't advance frame
pres@gsnPanelLabelBar = True
gsn_panel(wks,plot,(/2,1/),pres)
drawNDCGrid(wks)           ; Draws an NDC grid
frame(wks)                ; Now advance frame
```

```

pres          = True
pres@gsnFrame      = False
pres@gsnPanelLabelBar = True
pres@gsnPanelBottom    = 0.5
gsn_panel(wks,plot,(/2,1/),pres)


```



```

pres          = True
pres@gsnFrame      = False
pres@gsnPanelLabelBar = True
pres@gsnPanelLeft    = 0.5
gsn_panel(wks,plot,(/2,1/),pres)

```



```
;---Create two columns of panel plots
```

```
pres = True  
pres@gsnFrame = False  
pres@gsnPanelLabelBar = True
```

```
;---Left column
```


```
pres@gsnPanelLeft = 0.0  
pres@gsnPanelRight = 0.5  
gsn_panel(wks,plot(0::2),(/2,1/),pres)
```

```
;---Right column
```

```
pres@gsnPanelLeft = 0.5  
pres@gsnPanelRight = 1.0  
gsn_panel(wks,plot(1::2),(/2,1/),pres)
```


```
frame(wks)
```

Sea Level Pressure (millibars)



slp_two_panels.ncl

Sea Level Pressure (millibars)



```

;---Create one column of plots
pres                      = True
pres@gsnFrame              = False

;---Top 1/3 of NDC square
pres@gsnPanelTop           = 1.0
pres@gsnPanelBottom         = 0.6667
gsn_panel(wks,plot(0),(/1,1/),pres)

;---Bottom 2/3 of NDC square
pres@gsnPanelTop           = 0.6667
pres@gsnPanelBottom         = 0.0
gsn_panel(wks,plot(1:2),(/2,1/),pres)

;---Add the two custom labelbars
ncolors = dimsizes(res@cnLevels)+1
labels  = "" + res@cnLevels
colors1 = span_color_rgba(color_maps(0),ncolors)
colors2 = span_color_rgba(color_maps(1),ncolors)


lbres                      = True
lbres@vpWidthF             = 0.2
lbres@vpHeightF            = 0.3
lbres@lbOrientation         = "Vertical"
lbres@lbPerimOn             = False
lbres@lbLabelFontHeightF   = 0.012
lbres@lbLabelAlignment      = "InteriorEdges"
lbres@lbMonoFillPattern    = True
lbres@lbFillColors          = colors1

gsn_labelbar_ndc (wks,ncolors,labels,0.79,0.99,lbres)


;---Change location, size, colors of second labelbar
lbres@vpHeightF            = 0.55
lbres@lbFillColors          = colors2
gsn_labelbar_ndc (wks,ncolors,labels,0.79,0.62,lbres)

frame(wks)


```




slp_three_panels.ncl



```
pres@gsnPanelRowSpec = True  
gsn_panel(wks,plots,(/1,3,2,1/),pres
```





```

pres@gsnPanelRowSpec = True
pres@gsnPanelCenter = False
gsn_panel(wks,plots,(/1,3,2,1/),pres

```

Resources useful with gsn_panel	
gsnPanelLabelBar	Turn on panel labelbar
txString	Add title for paneled plots
gsnPanelFigureStrings	Add figure captions
gsnPanelScalePlotIndex	Index of plot to use for scaling all plots
gsnMaximize	Maximize paneled plots in frame
gsnPanelBottom / gsnPanelTop / gsnPanelLeft / gsnPanelRight	Control location of paneled plots in unit square
gsnPanelXWhiteSpacePercent / gsnPanelYWhiteSpacePercent	Control amount of white space between plots in a panel

Functions / procedures useful with gsn_panel	
gsn_create_labelbar	Create a custom labelbar
gsn_create_legend	Create a custom legend
gsn_legend_ndc	Draw a custom legend using NDC coordinates
gsn_labelbar_ndc	Draw a custom labelbar using NDC coordinates
drawNDCGrid	Draw NDC grid (for debugging purposes)

More advanced paneling

- See “panels” examples page:
NCL Home Page -> Examples -> *browser search* “panel”
<http://www.ncl.ucar.edu/Applications/panel.shtml>
- See list of every example that “**gsn_panel**” call appears in:
NCL Home Page -> Examples
*[List of tips, resources, **functions/procedures**, and the examples they appear in.]*
http://www.ncl.ucar.edu/Applications/func_list.shtml

Questions? ncl-talk@ucar.edu
<http://mailman.ucar.edu/mailman/listinfo/ncl-talk>



NCAR Mesa Lab
November 13, 2014